Techniques of Integration

Michael Penna, Indiana University – Purdue University, Indianapolis

Objective

If you just want to evaluate an integral then the `int` command in Maple can take you quite far with very little effort. But you may be interested in more than this: As you learn techniques of integration you'll be evaluating integrals by hand, and you may be interested in whether you're applying a technique properly. Or you may know you're applying a technique properly, but you may still be interested in finding an error in your work. In either case, Maple can again be very useful. In this project we illustrate how Maple can be used in learning techniques of integration. (Solutions manuals are nice, but they're static, not interactive. They may show you one way to evaluate an integral, but they may not tell you whether the way you'd like try is correct, and if it's not, why it's not.)

Narrative

The basic techniques of integration include u-substitution, integration by parts, techniques that apply to trig integrals, trig substitution, partial fractions, and other forms of substitution. Since space limits what we can say about these techniques, we refer you to your text for an in-depth discussion of these techniques, and illustrate how to perform them using Maple by example.

Note: The most important thing to get out of this project is some idea of how Maple commands are used: the specific order of the steps used is of secondary importance. Instead of focusing on (or thinking of) the sequence of steps in each task as you work through this project, focus on how the various commands get us from one place to another. (After all, the integrals evaluated in this project are so elementary that if you really wanted to evaluate them then you would probably be able to do so just as quickly and accurately by hand as you would using Maple.)

Tasks

1. Type the command lines in the left-hand column below into Maple in the order in which they are listed. These commands are aimed at using u-substitution to evaluate $\int_0^1 (x^3+1)e^{x^4+4x}\,dx$. Recall that the formula for u-substitution is:

$$\int_{x=a}^{b} f(u(x))u'(x)\;dx = \int_{u=u(a)}^{u(b)} f(u)\;du.$$

We begin by illustrating how quickly and easily Maple can evaluate this integral. We then use the `changevar` command in the `student` package to evaluate this integral by performing a u-substitution (in a way, slowing Maple down). The command `changevar(u=u(x),MyInt,u)` performs the u-substitution $u = u(x)$ in the integral Int.

`> # Your name, today´s date`	
`> # Techniques of Integration`	
`> restart;`	Clear Maple's memory.
`> with(student):`	Load the `student` package.
`> # Task 1:  u-substitution`	
`> MyInt := Int((x^3+1)*exp(x^4+4*x),x= 0..1);`	Here's our integral.
`> value(MyInt);`	Maple can quickly and easily evaluate many integrals.
`> changevar(u=x^4+4*x,MyInt,u);`	Use the u-substitution $u = x^4 + 4x$ to evaluate `MyInt`.
`> value(%);`	Finish evaluating the integral.

2. Continue by typing the command lines below into Maple. They are aimed at using integration by parts to evaluate $\int x^2 \sin x \; dx$. Recall that the formula for integration by parts is

$$\int u \; dv = uv - \int v \; du + C$$

and, as our example illustrates, this formula can be used more than once in evaluating an integral. Our computations involve the `intparts` command in the `student` package. The command `intparts(MyInt,u(x))` performs integration by parts on the integral $MyInt$ by letting $u = u(x)$.

`> # Task 2:  Parts`	
`> MyInt := Int(x^2*sin(x),x);`	Here's our integral.
`> intparts(MyInt,x^2);`	Integrate by parts, letting $u = x^2$.
`> intparts(%,x);`	Integrate by parts again, now letting $u = x$.
`> value(%);`	Finish evaluating the integral.

3. Continue by typing the command lines below into Maple. They are aimed at integrating $\int \cos^7 \theta \; d\theta$. Our computations involve both the `subs` command and the `powsubs` command in the `student` package. In this example, the `powsubs` command allows us to substitute "`1-sin(theta)^2`" for "`cos(theta)^2`".

`> # Task 3:  Trig integrals`	
`> MyInt := Int(cos(theta)^7,theta);`	Here's our integral.
`> powsubs(cos(theta)^2=1-sin(theta)^2,MyInt);`	
`> expand(%);`	
`> changevar(u=sin(theta),%,u);`	Do a u-substitution.
`> value(%);`	Evaluate this (simple) integral.
`> subs(u=sin(theta),%);`	Express the answer in terms of the original variable, `theta`.

In the above code we used `powsubs` to substitute into `MyInt`, we expanded `MyInt`, and then we evaluated. We could, however, just as easily used `powsubs` to substitute into the integrand, expand the integrand, and integrated the result as follows:

`> powsubs(cos(theta)^2=1-sin(theta)^2,cos(theta)^7);`	
`> expand(%);`	
`> Int(%,theta);`	
`> changevar(u=sin(theta),%,u);`	Do a u-substitution.
`> value(%);`	Evaluate this (simple) integral.
`> subs(u=sin(theta),%);`	Express the answer in terms of the original variable, `theta`.

(You do *not* have to type these last 6 lines of code into Maple.)

4. Continue by typing the command lines below into Maple. They are aimed at using a trig substitution to evaluate $\int_0^1 \sqrt{1-x^2} \; dx$. Note that our computations again involve the `changevar` command in the `student` package; now, however, we use this command in a slightly different way than we did in Task 1.

`> # Task 4:  Trig substitution`	
`> MyInt := Int(sqrt(1-x^2),x=0..1);`	Here's our integral.
`> changevar(x=sin(theta),MyInt,theta);`	Substitute `sin(theta)` for `x` in `MyInt`.
`> simplify(%);`	
`> powsubs(cos(theta)^2=(1+cos(2*theta))/2,%);`	
`> value(%);`	

5. Continue by typing the command lines below into Maple. They are aimed at using partial fractions to evaluate $\displaystyle\int \frac{1}{x^3 - x} \; dx$.

```
> # Task 5:  Partial Fractions
> f := x -> 1/(x^3-x);
> convert(f(x),parfrac,x);
```

Convert the expression `f(x)` in `x` into its partial fraction decomposition.

```
> Int(f(x),x) = Int(%,x);
```

Here's our integral.

```
> value(rhs(%));
```

At this time make a hard-copy of your typed input and Maple's responses. This hard-copy will be your lab report.

Comments

As indicated in the Narrative, in this project we admittedly limited our attention to integrating some fairly elementary functions. More complicated functions can easily require further manipulation; see Maple's Help for further commands and details on using them.